

# Examples of embedding Sage in L<sup>A</sup>T<sub>E</sub>X with SageT<sub>E</sub>X

Dan Drake and others

May 11, 2020

## 1 Inline Sage, code blocks

This is an example  $2 + 2 = 4$ . If you raise the current year mod 100 (which equals 20) to the power of the current day (11), you get 204800000000000. Also, 2020 modulo 42 is 4.

Code block which uses a variable `s` to store the solutions:

```
1+1
var('a,b,c')
eqn = [a+b*c==1, b-a*c==0, a+b==5]
s = solve(eqn, a,b,c)
```

Solutions of  $\text{eqn} = [bc + a = 1, -ac + b = 0, a + b = 5]$ :

$$\left[ a = -\frac{1}{4}i\sqrt{79} + \frac{11}{4}, b = \frac{1}{4}i\sqrt{79} + \frac{9}{4}, c = \frac{1}{10}i\sqrt{79} + \frac{1}{10} \right]$$
$$\left[ a = \frac{1}{4}i\sqrt{79} + \frac{11}{4}, b = -\frac{1}{4}i\sqrt{79} + \frac{9}{4}, c = -\frac{1}{10}i\sqrt{79} + \frac{1}{10} \right]$$

Now we evaluate the following block:

```
E = EllipticCurve("37a")
```

You can't do assignment inside `\sage` macros, since Sage doesn't know how to typeset the output of such a thing. So you have to use a code block. The elliptic curve  $E$  given by  $y^2 + y = x^3 - x$  has discriminant 37.

You can do anything in a code block that you can do in Sage and/or Python. Here we save an elliptic curve into a file.

```
try:
    E = load('E2')
except IOError:
    E = EllipticCurve([1,2,3,4,5])
    E.anlist(100000)
    E.save('E2')
```

The 9999th Fourier coefficient of  $y^2 + xy + 3y = x^3 + 2x^2 + 4x + 5$  is  $-27$ .


The following code block doesn't appear in the typeset file...but we can refer to whatever we did in that code block:  $e = 7$ .

```
var('x')
f(x) = log(sin(x)/x)
```

The Taylor Series of  $f$  begins:  $x \mapsto -\frac{1}{467775} x^{10} - \frac{1}{37800} x^8 - \frac{1}{2835} x^6 - \frac{1}{180} x^4 - \frac{1}{6} x^2$ .

## 2 Plotting


Here's a very large plot of the elliptic curve  $E$ .




You can use variables to hold plot objects and do stuff with them.

```
p = plot(f, x, -5, 5)
```

Here's a small plot of  $f$  from  $-5$  to  $5$ , which I've centered:



On second thought, use a size of  $\frac{3}{4}$  the `\textwidth` and don't use axes:




Remember, you're using Sage, and can therefore call upon any of the software packages Sage is built out of.

```
f = maxima('sin(x)^2*exp(x)')
g = f.integrate('x')
```


Plot  $g(x)$ , but don't typeset it.

You can specify a file format and options for `includegraphics`. The default is for EPS and PDF files, which are the best choice in almost all situations. (Although see the section on 3D plotting.)



If you use regular `latex` to make a DVI file, you'll see a box, because DVI files can't include PNG files. If you use `pdflatex` that will work. See the documentation for details.


When using `\sageplot`, you can pass in just about anything that Sage can call `.save()` on to produce a graphics file:



To fiddle with aspect ratio, first save the plot object:

```
p = plot(x, 0, 1) + circle((0,0), 1)
p.set_aspect_ratio(1)
```

Now plot it and see the circular circle and nice 45 degree angle:




Indentation and so on works fine.

```
s      = 7
s2     = 2^s
P.<x>  = GF(2) []
M      = matrix(parent(x),s2)
for i in range(s2):
    p = (1+x)^i
    pc = p.coefficients(sparse=False)
    a = pc.count(1)
    for j in range(a):
        idx = pc.index(1)
        M[i,idx+j] = pc.pop(idx)

matrixprogram = matrix_plot(M,cmap='Greys')
```

And here's the picture:



Reset `x` in Sage so that it's not a generator for the polynomial ring: `x`

## 2.1 Plotting (combinatorial) graphs with TikZ

Sage now includes some nice support for plotting graphs using TikZ. Here, we mean things with vertices and edges, not graphs of a function of one or two variables.


The graphics in this section depends on the `tkz-berge` package, which is generally only available in newer  $\text{\TeX}$  distributions (for example,  $\text{\TeX}$ Live 2011 and newer). That package depends in turn on TikZ 2.0, which is also only available in newer  $\text{\TeX}$  distributions. Installing both of those is in some cases nontrivial, so this section is disabled by default.

If you have TikZ and `tkz-berge` and friends, remove the `comment` environments below.

## 2.2 3D plotting


3D plotting right now (Sage version 4.3.4) is problematic because there's no convenient way to produce vector graphics. We can make PNGs, though, so if you pass `sageplot` a graphics object that cannot be saved to EPS or PDF format, we will automatically save to a PNG file, which can be used when typesetting a PDF file, but not when creating a DVI file. However, you can specify the `"imagemagick"` option, which will use the `ImageMagick convert` utility to make EPS files. See the documentation for details.

Here's a 3D plot whose format we do not specify; it will automatically get saved as a PNG file and won't work when using `latex` to make a DVI file.



Here's the (perhaps-not-so-) famous Sage cube graph in 3D.

```
G = graphs.CubeGraph(5)
```



### 3 Pausing SageTeX

Sometimes you want to “pause” for a bit while writing your document if you have embedded a long calculation or just want to concentrate on the  $\text{\LaTeX}$  and ignore any Sage stuff. You can use the `\sagetexpause` and `\sagetexunpause`

macros to do that.

A calculation: (SageTeX is paused) and a code environment that simulates a time-consuming calculation. While paused, this will get skipped over.

```
import time
time.sleep(15)
```

Graphics are also skipped: SageTeX is paused; no graphic

## 4 Make Sage write your L<sup>A</sup>T<sub>E</sub>X for you

With SageTeX, you can not only have Sage do your math for you, it can write parts of your L<sup>A</sup>T<sub>E</sub>X document for you! For example, I hate writing `tabular` environments; there's too many fiddly little bits of punctuation and whatnot... and what if you want to add a column? It's a pain—or rather, it *was* a pain. Just write a Sage/Python function that outputs a string of L<sup>A</sup>T<sub>E</sub>X code, and use `\sagestr`. Here's how to make Pascal's triangle.

```
def pascals_triangle(n):
    # start of the table
    s = [r"\begin{tabular}{cc|" + "r" * (n+1) + "|}"]
    s.append(r" & & $k$: & \\")
    # second row, with k values:
    s.append(r" & ")
    for k in [0..n]:
        s.append("& {0} ".format(k))
    s.append(r"\\")
    # the n = 0 row:
    s.append(r"\hline" + "\n" + r"$n$: & 0 & 1 & \\")
    # now the rest of the rows
    for r in [1..n]:
        s.append(" & {0} ".format(r))
        for k in [0..r]:
            s.append("& {0} ".format(binomial(r, k)))
        s.append(r"\\")
    # add the last line and return
    s.append(r"\end{tabular}")
    return ''.join(s)

# how big should the table be?
n = 8
```

Okay, now here's the table. To change the size, edit `n` above. If you have several tables, you can use this to get them all the same size, while changing only one thing.

		<i>k</i> :									
		0	1	2	3	4	5	6	7	8	
<i>n</i> :	0	1									
	1	1	1								
	2	1	2	1							
	3	1	3	3	1						
	4	1	4	6	4	1					
	5	1	5	10	10	5	1				
	6	1	6	15	20	15	6	1			
	7	1	7	21	35	35	21	7	1		
	8	1	8	28	56	70	56	28	8	1	

## 5 Include doctest-like examples in your document

Here are some examples of using the `sageexample` environment:

```
sage: 2+2
```

4

```
sage: print('middle')
```

None

```
sage: factor(x^2 + 2*x + 1)
```

$(x + 1)^2$

Note above that no output from the `print` statement appears. That is because we have to use Python's `exec` to execute that statement (and not `eval()`), and we can't get the output from that.

That said, if you want to see the plain-text output you put into your `.tex` file as well as the Sage-computed typeset output, renew the `sageexampleincludetextoutput` command to `True`:

```
\renewcommand{\sageexampleincludetextoutput}{True}
```

This can be useful to check that the two outputs are consistent. Here's the `print` statement with text output included:

```
sage: print('middle')
```

```
middle
```

None

When typesetting your document, the validity of the outputs is not checked. In fact, the provided outputs are completely ignored:

```
sage: is_prime(57)
```

```
toothpaste
```

False

Multiline statements with the “...:” continuation marks are supported, as are triple-quoted strings delimited by single quotes (double quotes won’t work):

```
sage: gcd([5656565656,  
....:    4747474747,  
....:    123456789])
```

1

```
sage: mystr = '''my  
....: string  
....: has  
....: several  
....: lines.'''
```

```
sage: len(mystr)
```

28

```
sage: def f(a):  
....:     '''This function is really quite nice,  
....:     although perhaps not very useful.'''  
....:     print("f called with a = {}".format(a))  
....:     y = integrate(SR(cyclotomic_polynomial(10)) + a, x)  
....:     return y + 1
```

```
sage: f(x)
```

$$\frac{1}{5}x^5 - \frac{1}{4}x^4 + \frac{1}{3}x^3 + x + 1$$

Note that the “*f* called with...” stuff doesn’t get typeset, since when running Sage on `example.sagetex.sage`, that gets printed to the terminal.

Typesetting your document produces a file named `example_doctest.sage` containing all the doctest-like examples, and you can have Sage check them for you with:

```
$ sage -t example_doctest.sage
```

You should get a doctest failure from the “toothpaste” line above. The line numbers from `sage -t` refer to the “`_doctest.sage`” file.

Beware that `sage -t` does not really handle file names with special characters in them, particularly dashes, dots, and spaces—this ultimately comes from the way Python interprets `import` statements. Also, running doctests on files outside the main Sage library does not always work, so contact `sage-support` if you run into troubles.

Some more examples. This environment is implemented a little bit differently than the other environments, so it’s good to make sure that definitions are preserved across multiple uses. This will correctly define *a*, but not print its output because the statement is made up of a sequence of expressions and we can’t use Python’s `eval()`; we have to use `exec` and we can’t capture the output from that.

```
sage: 1; 2; a=4; 3; a
```

However, after that, Sage should remember that  $a = 4$  and be able to use that in future `sageexample` blocks:

```
sage: f(a)
```

$$\frac{1}{5}x^5 - \frac{1}{4}x^4 + \frac{1}{3}x^3 - \frac{1}{2}x^2 + 5x + 1$$

## 6 Plotting functions in TikZ with SageTeX

(The code in this section should work with any reasonable version of TikZ, which means it should work with all but the most terribly out-of-date TeX installations—but to make sure we can accommodate everyone, the code here is commented out. You can almost certainly uncomment and run them. Make sure you do `\usepackage{tikz}` in the preamble.)

## 7 The sagecommandline environment

When writing a TeX document about Sage, you may want to show some examples of commands and their output. But naturally, you are lazy and don't want to cut and paste the output into your document. “Why should I have to do anything? Why can't Sage and TeX cooperate and do it for me?” you may cry. Well, they *can* cooperate:

```
sage: 1+1 1
2 2
sage: is_prime(57) 3
False 4
sage: if is_prime(57): 5
.....:     print('prime') 6
.....: else: 7
.....:     print('composite') 8
```

Note that the output of the commands is not included in the source file, but are included in the typeset output.

Because of the way the environment is implemented, not everything is exactly like using Sage in a terminal: the two commands below (and the “if is prime” one above, did you notice that?) would produce some output, but don't here:

```
sage: x = 2010; len(x.divisors()) 9
sage: print('Hola, mundo!') 10
None 11
```

The difference lies in the Python distinction between statements and expressions; we can use `eval()` for an expression and get its output, but we must use `exec` for a statement and can't get the output, if any.

One nice thing is that you can set labels by using an @ sign:

```
sage: l = matrix([[1,0,0],[3/5,1,0],[-2/5,-2,1]]) 12
```

```

sage: d = diagonal_matrix([15, -1, 4]) 13
sage: u = matrix([[1,0,1/3],[0,1,2],[0,0,1]]) # foo 14
sage: l*d*u # this is a comment 15
[15 0 5] 16
[ 9 -1 1] 17
[-6 2 6] 18

```

And then refer to that label: it was on line 13, which is on page 12. Note that the other text after the hash mark on that line does not get typeset as a comment, and that you cannot have any space between the hash mark and the @. You will also need to typeset your document *twice*

You can also typeset the output by changing the value of `\sagecommandlinetextoutput` to False:

```

sage: l*d*u 19

```

$$\begin{pmatrix} 15 & 0 & 5 \\ 9 & -1 & 1 \\ -6 & 2 & 6 \end{pmatrix}$$

```

sage: x = var('x') 20
sage: (1-cos(x)^2).trig_simplify() 21

```

$$\sin(x)^2$$

The Sage input and output is typeset using the `listings` package with the styles `SageInput` and `SageOutput`, respectively. If you don't like the defaults you can change them. It is recommended to derive from `DefaultSageInput` and `DefaultSageOutput`, for example...makes things overly colorful:

```

sage: pi.n(100) 22
3.1415926535897932384626433833 23

```

Plotting things doesn't automatically pull in the plot, just the text representation of the plot (the equivalent of applying `str()` to it):

```

sage: plot(sin(x), (x, 0, 2*pi)) 24
Graphics object consisting of 1 graphics primitive 25

```

You can include output, but it will be ignored. This is useful for doctesting, as all the `sagecommandline` environment things get put into the `“.doctest.sage”` file. However, note that if you don't include any output, then the corresponding doctest will fail if the command produces output. The doctest output from this file will have lots of failures because not many of the commands have output included in the source `.tex` file.

The command below has incorrect output included in the `.tex` file; in the PDF, you see the correct Sage-computed answer, but if you do `sage -t example_doctest.sage` you will get a genuine doctest failure.

```

sage: factor(x^2 + 2*x + 1) 26
(x + 1)^2 27

```